```
/**
 * CS 240 Connect Four: colcheck.s
 * Implement this function in assembly code.
 * Check if the board grid contains four connected pieces of type p
 * in a column, including the piece p just played at [row,col].
 *
 * Simpler, less efficient algorithm.
 */
/*
long checkcol(piece grid[6][7], long row, long col, piece p) {
  long i;
  for (i = 0; i <= row; i++) {
    if (grid[row - i][col] != p) {
      return 0;
    }
  }
  return i >= 4;
}
*/
    .text
    .align 8
    .globl checkcol

// long checkcol(piece grid[6][7], long row, long col, piece p);
checkcol:
    // grid = %rdi
    // row = %rsi
    // col = %rdx
    // p = %rcx
    // i = %r10
    // i = 0;
    movq $0, %r10
loopTop:
    // test i <= row ?  (a.k.a. if NOT i > row)
    cmpq %rsi, %r10
    jg loopEnd
    // get row - i   // REGISTER decision caller/callee
     // row
    movq %rsi, %r8
     // row - i
    subq %r10, %r8
      // scale row index by #cols
    imulq $7, %r8
    // add col
    addq %rdx, %r8
    // get grid[(row - i)*COLS + col]
     // element from grid
    movq (%rdi, %r8, 8),%r9
```

```
49      // if (element != p)
50      cmpq %r9, %rcx
51      je endIf
52      // return 0;
53      movq $0, %rax
54      jmp end
55  endIf:
56       // i ++
57      incq %r10
58      jmp loopTop
59  loopEnd:
60      // return i >= 4
61      movq $0, %rax
62      movq $1, %r11
63      cmpq $4, %r10
64      cmovge %r11, %rax
65  end:
66      retq
67
```